
10 Policy Search

Policy search involves searching the space of policies without directly computing
a value function. The policy space is often lower-dimensional than the state
space and can often be searched more efficiently. Policy optimization optimizes
the parameters in a parameterized policy to maximize utility. This parameterized
policy can take many different forms, such as neural networks, decision trees,
and computer programs. This chapter begins by discussing a way to estimate the
value of a policy given an initial state distribution. We will then discuss search
methods that do not use estimates of the gradient of the policy, saving gradient
methods for the next chapter. Although local search can be quite effective in
practice, we will discuss a few alternative optimization approaches that can avoid
local optima.1 1 There are many other optimiza-

tion approaches, as discussed
by M. J. Kochenderfer and
T.A. Wheeler, Algorithms for
Optimization. MIT Press, 2019.

10.1 Approximate Policy Evaluation

As introduced in section 7.2, we can compute the expected discounted return
when following a policy π from a state s. This expected discounted return Uπ(s)
can be computed iteratively (algorithm 7.3) or through matrix operations (algo-
rithm 7.4), when the state space is discrete and relatively small. We can use these
results to compute the expected discounted return of π:

U(π) = ∑
s

b(s)Uπ(s) (10.1)

assuming an initial state distribution b(s).
We will use this definition of U(π) throughout this chapter. However, we often

cannot compute U(π) exactly when the state space is large or continuous. Instead,
we can approximate U(π) by sampling trajectories, consisting of states, actions,

and rewards when following π. The definition of U(π) can be rewritten

U(π) = Eτ [R(τ)] =
∫

pπ(τ)R(τ)dτ (10.2)

where pπ(τ) is the probability density associatedwith trajectory τ when following
policy π starting from initial state distribution b. The trajectory reward R(τ) is the
discounted return associated with τ. Figure 10.1 illustrates the computation of
U(π) in terms of trajectories sampled from an initial state distribution.

U(π)

Figure 10.1. The utility associated
with a policy from an initial state
distribution is computed from the
return associated with all possible
trajectories under the given policy,
weighted according to their likeli-
hood.

Monte Carlo policy evaluation (algorithm 10.1) involves approximating equa-
tion (10.2) with m trajectory rollouts of π:

U(π) ≈ 1
m

m

∑
i=1

R(τ(i)) (10.3)

where τ(i) is the ith trajectory sample.

struct MonteCarloPolicyEvaluation

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

end

function (U::MonteCarloPolicyEvaluation)(π)

R(π) = rollout(U.𝒫, rand(U.b), π, U.d)

return mean(R(π) for i = 1:U.m)

end

(U::MonteCarloPolicyEvaluation)(π, θ) = U(s->π(θ, s))

Algorithm 10.1. Monte Carlo pol-
icy evaluation of a policy π. The
method runs m rollouts to depth d

according to the dynamics speci-
fied by the problem 𝒫. Each rollout
is run from an initial state sampled
from state distribution b. The final
line in this algorithm block evalu-
ates a policy π parameterized by
θ, which will be useful in the algo-
rithms in this chapter that attempt
to find a θ that maximizes U.

Monte Carlo policy evaluation is stochastic. Multiple evaluations of equa-
tion (10.1) with the same policy can give different estimates. Increasing the
number of rollouts decreases the variance of the evaluation, as demonstrated in
figure 10.2.

Figure 10.2. The effect of depth
and sample count for Monte Carlo
policy evaluation of a uniform ran-
dom policy on the cart-pole prob-
lem (appendix F.3). The variance
decreases as the number of sam-
ples increases. The blue regions in-
dicate the 5 % to 95 % and 25 % to
75 % empirical quantiles of U(π).

We will use πθ to denote a policy parameterized by θ. For convenience, we will
use U(θ) as shorthand for U(πθ) in cases where it is not ambiguous. The param-
eter θ may be a vector or some other more complex representation. For example,
we may want to represent our policy using a neural network with a particular
structure. We would use θ to represent the weights in the network. Many opti-
mization algorithms assume that θ is a vector with a fixed number of components.

Other optimization algorithms allow for more flexible representations, including
representations like decision trees or computational expressions.2 2 We will not be discussing them

here, but some are implemented in
ExprOptimization.jl.

10.2 Local Search

A common approach to optimization is local search, where we begin with an initial
parameterization and incrementallymove from neighbor to neighbor in the search
space until convergence. We discussed this type of approach in chapter 5 in the
context of optimizing Bayesian network structures with respect to the Bayesian
score. Here, we are optimizing policies parameterized by θ. We are trying to find
a θ that maximizes U(θ).

There are many different local search algorithms, but this section will focus
on the Hooke-Jeeves method (algorithm 10.2).3 This algorithm assumes that our 3 R. Hooke and T.A. Jeeves, ‘‘Direct

Search Solution of Numerical and
Statistical Problems,’’ Journal of the
ACM (JACM), vol. 8, no. 2, pp. 212–
229, 1961.

policy is parameterized by an n-dimensional vector θ. The algorithm takes a
step of size ±α in each of the coordinate directions from the current θ. These 2n
points correspond to the neighborhood of θ. If no improvements to the policy
are found, then the step size α is decreased by some factor. If an improvement
is found, it moves to the best point. The process continues until α drops below
some threshold ε > 0. Figure 10.3 illustrates this process.

10.3 Genetic Algorithms

A potential issue with local search algorithms like the Hooke-Jeeves method is
that the optimization can get stuck in a local optimum. There are a wide variety of
approaches that involve maintaining a population consisting of samples of points
in the parameter space, evaluating them in parallel with respect to our objective,
and then recombining them in some way to drive the population towards a global
optimum. A genetic algorithm4 is one such approach, which derives inspiration

4 D. E. Goldberg,Genetic Algorithms
in Search, Optimization, and Machine
Learning. Addison-Wesley, 1989.

from biological evolution. It is a general optimization method, but it has been
successful in the context of optimizing policies. For example, this approach has
been used to optimize policies for Atari video games, where the policy parameters
correspond to weights in a neural network.5

5 F. P. Such, V. Madhavan, E. Conti,
J. Lehman, K.O. Stanley, and J.
Clune, ‘‘Deep Neuroevolution: Ge-
netic Algorithms Are a Competi-
tive Alternative for Training Deep
Neural Networks for Reinforce-
ment Learning,’’ 2017. arXiv: 1712
.06567v3. The implementation in
this section follows their relatively
simple formulation. Their formu-
lation does not include crossover,
which is typically used to mix pa-
rameterizations across a popula-
tion.

A simple version of this approach (algorithm 10.3) begins with a population of
m random parameterizations, θ(1), . . . ,θ(m). We compute U(θ(i)) for each sample
i in the population. Since these evaluations potentially involve many rollout

2021-01-07 22:28:48-08:00, revision abe81c1, draft: send comments to bugs@algorithmsbook.com

https://arxiv.org/abs/1712.06567v3
https://arxiv.org/abs/1712.06567v3


212 chapter 10. policy search

struct HookeJeevesPolicySearch

θ # initial parameterization

α # step size

c # step size reduction factor

ϵ # termination step size

end

function optimize(M::HookeJeevesPolicySearch, π, U)

θ, θ′, α, c, ϵ = copy(M.θ), similar(M.θ), M.α, M.c, M.ϵ

u, n = U(π, θ), length(θ)

while α > ϵ

copyto!(θ′, θ)

best = (i=0, sgn=0, u=u)

for i in 1:n

for sgn in (-1,1)

θ′[i] = θ[i] + sgn*α

u′ = U(π, θ′)

if u′ > best.u

best = (i=i, sgn=sgn, u=u′)

end

end

θ′[i] = θ[i]

end

if best.i != 0

θ[best.i] += best.sgn*α

u = best.u

else

α *= c

end

end

return θ

end

Algorithm 10.2. Policy search
using the Hooke-Jeeves method,
which returns a θ that has been op-
timized with respect to U. The pol-
icy π takes as input a parameter θ
and state s. This implementation
startswith an initial θ. The step size
α is reduced by a factor of c if no
neighbor improves the objective. It-
erations are run until the step size
is less than ϵ.

10.3. genetic algorithms 213

Suppose we want to optimize a policy for the simple regulator problem
described in appendix F.5. We define a stochastic policy π parameterized by
θ such that the action is generated according to

a ∼ N (θ1s, (|θ2|+ 10−5)2) (10.4)

The following code defines the parameterized stochastic policy π, evaluation
function U, and method M. It then calls optimize(M, π, U), which returns
an optimized value for θ. In this case, we use the Hooke-Jeeves method, but
the other methods discussed in this chapter can be passed in as M instead.

function π(θ, s)

return rand(Normal(θ[1]*s, abs(θ[2]) + 0.00001))

end

b, d, n_rollouts = Normal(0.3,0.1), 10, 3

U = MonteCarloPolicyEvaluation(𝒫, b, d, n_rollouts)

θ, α, c, ϵ = [0.0,1.0], 0.75, 0.75, 0.01

M = HookeJeevesPolicySearch(θ, α, c, ϵ)

θ = optimize(M, π, U)

Example 10.1. An example of using
a policy optimization algorithm
to optimize the parameters of a
stochastic policy.

Figure 10.3. The Hooke-Jeeves
method applied to optimizing a
policy in the simple regulator prob-
lem discussed in example 10.1. The
evaluations at each iteration are
shown as white points. Iterations
proceed left to right and top to bot-
tom, and the background is colored
according to the expected utility.

214 chapter 10. policy search

simulations and are therefore computationally expensive, they are often run in
parallel. These evaluations help us identify the elite samples, which are the top
melite samples according to U.

The population at the next iteration is generated by producing m− 1 new pa-
rameterizations by repeatedly selecting a random elite sample θ and perturbing it
with isotropic Gaussian noise, θ+ σε, where ε ∼ N (0, I). The best parameteriza-
tion, unperturbed, is included as the mth sample. Because the evaluations involve
stochastic rollouts, a variation of this algorithm could involve running additional
rollouts to help identify which of the elite samples is truly the best. Figure 10.4
shows several iterations, or generations, of this approach on an example problem.

struct GeneticPolicySearch

θs # initial population

σ # initial standard devidation

m_elite # number of elite samples

k_max # number of iterations

end

function optimize(M::GeneticPolicySearch, π, U)

θs, σ = M.θs, M.σ

n, m = length(first(θs)), length(θs)

for k in 1:M.k_max

us = [U(π, θ) for θ in θs]

sp = sortperm(us, rev=true)

θ_best = θs[sp[1]]

rand_elite() = θs[sp[rand(1:M.m_elite)]]

θs = [rand_elite() + σ.*randn(n) for i in 1:(m-1)]

push!(θs, θ_best)

end

return last(θs)

end

Algorithm 10.3. A genetic policy
search method for iteratively up-
dating a population of policy pa-
rameterizations θs, which takes a
policy evaluation function U, a pol-
icy π(θ, s), a perturbation stan-
dard deviation σ, an elite sample
count m_elite, and an iteration
count k_max. The best m_elite sam-
ples from each iteration are used to
generate the samples for the subse-
quent iteration.

Figure 10.4. Genetic policy search
with σ = 0.25 applied to the simple
regulator problem using 25 sam-
ples per iteration. The five elite
samples in each generation are
shown in red, with the best sam-
ple drawn with a larger dot.

10.4 Cross Entropy Method

The cross entropy method (algorithm 10.4) involves updating a search distribution
over the parameterized space of policies at each iteration.6 We parameterize this 6 S. Mannor, R. Y. Rubinstein, and Y.

Gat, ‘‘The Cross Entropy Method
for Fast Policy Search,’’ in Interna-
tional Conference on Machine Learn-
ing (ICML), 2003.

search distribution p(θ | ψ) with ψ.7 This distribution can belong to any family,

7 Often, θ and ψ are vectors, but
because this assumption is not re-
quired for this method, we will not
bold them in this section.

but a Gaussian distribution is a common choice, where ψ represents the mean
and covariance of the distribution. The objective is to find a ψ∗ that maximizes
the expectation of U(θ) when θ is drawn from the search distribution:

ψ∗ = arg max
ψ

E
θ∼p(·|ψ)

[U(θ)] = arg max
ψ

∫
θ

U(θ)p(θ | ψ)dθ (10.5)

Directly maximizing equation (10.5) is typically computationally infeasible.
The approach taken in the cross entropy method is to start with an initial ψ,
typically chosen so that the distribution is spread over the relevant parameter
space. At each iteration, we draw m samples from the associated distribution and
then update ψ to fit the elite samples. For the fit, we typically use the maximum
likelihood estimate (section 4.1).8 We stop after some fixed number of iterations or

8 The maximum likelihood esti-
mate corresponds to the choice of
ψ that minimizes the cross entropy
(see appendix A.9) between the
search distribution and the elite
samples.

until the search distribution becomes highly focused on an optimum. Figure 10.5
demonstrates the algorithm on a simple problem.

struct CrossEntropyPolicySearch

p # initial distribution

m # number of samples

m_elite # number of elite samples

k_max # number of iterations

end

function optimize_dist(M::CrossEntropyPolicySearch, π, U)

p, m, m_elite, k_max = M.p, M.m, M.m_elite, M.k_max

for k in 1:k_max

θs = rand(p, m)

us = [U(π, θs[:,i]) for i in 1:m]

θ_elite = θs[:,sortperm(us)[(m-m_elite+1):m]]

p = Distributions.fit(typeof(p), θ_elite)

end

return p

end

function optimize(M, π, U)

return Distributions.mode(optimize_dist(M, π, U))

end

Algorithm 10.4. Cross entropy pol-
icy search, which iteratively im-
proves a search distribution ini-
tially set to p. This algorithm takes
as input a parameterized policy
π(θ, s) and a policy evaluation
function U. In each iteration, m

samples are drawn and the top
m_elite are used to refit the dis-
tribution. The algorithm termi-
nates after k_max iterations. The
distribution p can be defined us-
ing the Distributions.jl pack-
age. For example, we might define
μ = [0.0,1.0]

Σ = [1.0 0.0; 0.0 1.0]

p = MvNormal(μ,Σ)

Figure 10.5. The cross entropy
method applied to the simple regu-
lator problem using a multivariate
Gaussian search distribution. The
five elite samples in each iteration
are shown in red. The initial distri-
bution is set to N ([0, 3], 2I).

10.5 Evolution Strategies

Evolution strategies9 update a search distribution parameterized by a vector ψ at 9 D. Wierstra, T. Schaul, T. Glas-
machers, Y. Sun, J. Peters, and
J. Schmidhuber, ‘‘Natural Evolu-
tion Strategies,’’ Journal of Machine
Learning Research, 2014.

each iteration. However, instead of fitting the distribution to a set of elite samples,
it updates the distribution by taking a step in the direction of the gradient.10 The

10 We are effectively doing gradi-
ent ascent, which is reviewed in ap-
pendix A.11.

gradient of the objective in equation (10.5) can be computed as follows:11

11 The policy parameter θ is not
bolded here because it is not re-
quired to be a vector. However, ψ
is in bold because we require it to
be a vector when we work with the
gradient of the objective.

∇ψ E
θ∼p(·|ψ)

[U(θ)] = ∇ψ
∫

U(θ)p(θ | ψ)dθ (10.6)

=
∫

U(θ)∇ψp(θ | ψ)dθ (10.7)

=
∫

U(θ)∇ψp(θ | ψ) p(θ | ψ)
p(θ | ψ) dθ (10.8)

=
∫ (

U(θ)∇ψ log p(θ | ψ)
)

p(θ | ψ)dθ (10.9)

= E
θ∼p(·|ψ)

[
U(θ)∇ψ log p(θ | ψ)

]
(10.10)

This computation requires knowing the likelihood gradient ∇ψ log p(θ | ψ).
However, we can often compute this analytically as discussed in example 10.2.

The search gradient can be estimated from m samples: θ(1), . . . , θ(m) ∼ p(· | ψ):

∇ψ E
θ∼p(·|ψ)

[U(θ)] ≈ 1
m

m

∑
i=1

U(θ(i))∇ψ log p(θ(i) | ψ) (10.11)

This search gradient estimate depends on the evaluated expected utility, which
itself tends to vary widely. We can make our gradient estimate more resilient
with rank shaping, which replaces the utility values with weights based on the
relative performance of each sample to the other samples in its iteration. The m
samples are sorted in descending order of expected utility. Weight w(i) is assigned

to sample i according to some weighting scheme with w(1) ≥ · · · ≥ w(m). The
search gradient becomes:

∇ψ E
θ∼p(·|ψ)

[U(θ)] ≈
m

∑
i=1

w(i)∇ψ log p(θ(i) | ψ) (10.12)

A common weighting scheme is:12 12 N. Hansen and A. Ostermeier,
‘‘Adapting Arbitrary Normal Mu-
tation Distributions in Evolution
Strategies: the Covariance Matrix
Adaptation,’’ in IEEE International
Conference on Evolutionary Computa-
tion, 1996.

w(i) =
max

(
0, log

(m
2 + 1

)
− log(i)

)
∑m

j=1 max
(
0, log

(m
2 + 1

)
− log(j)

) − 1
m

(10.13)

These weights, shown in figure 10.6, favor better samples and give most samples
a small negative weight. Rank-shaping reduces the influence of outliers in each
iteration.

Figure 10.6. Several weight-
ings constructed using equa-
tion (10.13).

Algorithm 10.5 provides an implementation of the evolution strategies method.
Figure 10.7 shows an example search progression.

10.6 Isotropic Evolutionary Strategies

The previous section introduced evolutionary strategies that can work with gen-
eral search distributions. This section will make the assumption that the search
distribution is a spherical or isotropic Gaussian, where the covariance matrix of
the form σ2I.13 Under this assumption, the expected utility of the distribution

13 An example of this approach ap-
plied to policy search is explored
by T. Salimans, J. Ho, X. Chen, S.
Sidor, and I. Sutskever, ‘‘Evolution
Strategies as a Scalable Alternative
to Reinforcement Learning,’’ 2017.
arXiv: 1703.03864v2.

introduced in equation (10.5) simplifies to:14

14 In general, if A>A = Σ, thenθ =
µ + A>ε transforms ε ∼ N (0, I)
into a sample θ ∼ N (µ, Σ).

struct EvolutionStrategies

D # distribution constructor

ψ # initial distribution parameterization

∇logp # log search likelihood gradient

m # number of samples

α # step factor

k_max # number of iterations

end

function evolution_strategy_weights(m)

ws = [max(0, log(m/2+1) - log(i)) for i in 1:m]

ws ./= sum(ws)

ws .-= 1/m

return ws

end

function optimize_dist(M::EvolutionStrategies, π, U)

D, ψ, m, ∇logp, α = M.D, M.ψ, M.m, M.∇logp, M.α

ws = evolution_strategy_weights(m)

for k in 1:M.k_max

θs = rand(D(ψ), m)

us = [U(π, θs[:,i]) for i in 1:m]

sp = sortperm(us, rev=true)

∇ = sum(w.*∇logp(ψ, θs[:,i]) for (w,i) in zip(ws,sp))

ψ += α.*∇

end

return D(ψ)

end

Algorithm 10.5. An evolution
strategies method for updating a
search distribution D(ψ) over pol-
icy parameterizations for policy
π(θ, s). This implementation also
takes an initial search distribution
parameterization ψ, the log search
likelihood gradient ∇logp(ψ, θ),
a policy evaluation function U,
and an iteration count k_max. In
each iteration, m parameteriza-
tion samples are drawn and are
used to estimate the search gradi-
ent. This gradient is then applied
with a step factor α. We can use
Distributions.jl to define D(ψ).
For example, if we want to define
D to construct a Gaussian with a
given mean ψ and fixed covariance
Σ, we can use
D(ψ) = MvNormal(ψ, Σ).

−2 0 2

1

3

5

θ1

θ 2

−2 0 2

θ1

−2 0 2

θ1

−2 0 2

θ1

Figure 10.7. Evolution strategies,
algorithm 10.5, applied to the sim-
ple regulator problem using a mul-
tivariate Gaussian search distribu-
tion. Samples are shown in white,
along with their search gradient
contributions w∇ log p.

The multivariate normal distribution N (µ, Σ) with mean µ and covariance
Σ is a common distribution family. The likelihood in d dimensions has the
form

p(x | µ, Σ) = (2π)−
d
2 |Σ|−

1
2 exp

(
−1

2
(x− µ)>Σ−1(x− µ)

)
where |Σ| is the determinant of Σ. The log likelihood is

log p(x | µ, Σ) = −d
2

log(2π)− 1
2

log |Σ| − 1
2
(x− µ)>Σ−1(x− µ)

The parameters can be updated using their log likelihood gradients:

∇µ log p(x | µ, Σ) = Σ−1(x− µ)

∇Σ log p(x | µ, Σ) =
1
2

Σ−1(x− µ)(x− µ)>Σ−1 − 1
2

Σ−1

The term ∇Σ contains the partial derivative of each entry of Σ with respect
to the log likelihood.

Directly updating Σ may not result in a positive definite matrix, as is re-
quired for covariance matrices. One solution is to represent Σ as a product
A>A, which guarantees that Σ remains positive semidefinite, and then up-
date A instead. Replacing Σ by A>A and taking the gradient with respect to
A yields:

∇(A) log p(x | µ, A) = A
[
∇Σ log p(x | µ, Σ) +∇Σ log p(x | µ, Σ)>

]

Example 10.2. A derivation of the
log likelihood gradient equations
for the multivariate Gaussian dis-
tribution. For the original deriva-
tion and severalmore sophisticated
solutions for handling the positive
definite covariance matrix, see D.
Wierstra, T. Schaul, T. Glasmach-
ers, Y. Sun, J. Peters, and J. Schmid-
huber, ‘‘Natural Evolution Strate-
gies,’’ Journal of Machine Learning
Research, 2014.

E
θ∼N (ψ,σ2I)

[U(θ)] = E
ε∼N (0,I)

[U(ψ+ σε)] (10.14)

The search gradient reduces to:

∇ψ E
θ∼N (ψ,σ2I)

[U(θ)] = E
θ∼N (ψ,σ2I)

[
U(θ)∇ψ log p(θ | ψ, σ2I)

]
(10.15)

= E
θ∼N (ψ,σ2I)

[
U(θ)

1
σ2 (θ−ψ)

]
(10.16)

= E
ε∼N (0,I)

[
U(ψ+ σε)

1
σ2 (σε)

]
(10.17)

=
1
σ

E
ε∼N (0,I)

[U(ψ+ σε)ε] (10.18)

Algorithm 10.6 provides an implementation of this strategy. This implemen-
tation incorporates mirrored sampling.15 We sample m/2 values from the search

15 D. Brockhoff, A. Auger, N.
Hansen, D. Arnold, and T. Hohm,
‘‘Mirrored Sampling and Se-
quential Selection for Evolution
Strategies,’’ in International Confer-
ence on Parallel Problem Solving from
Nature, 2010.

distribution and then generate the other m/2 samples by mirroring them about
the mean. Mirrored samples reduce the variance of the gradient estimate.16 The

16 This technique was implemented
by T. Salimans, J. Ho, X. Chen, S.
Sidor, and I. Sutskever, ‘‘Evolution
Strategies as a Scalable Alterna-
tive to Reinforcement Learning,’’
2017. arXiv: 1703.03864v2. They
included other techniques as well,
including weight decay.benefit of this technique is shown in figure 10.8.

struct IsotropicEvolutionStrategies

ψ # initial mean

σ # initial standard devidation

m # number of samples

α # step factor

k_max # number of iterations

end

function optimize_dist(M::IsotropicEvolutionStrategies, π, U)

ψ, σ, m, α, k_max = M.ψ, M.σ, M.m, M.α, M.k_max

n = length(ψ)

ws = evolution_strategy_weights(2*div(m,2))

for k in 1:k_max

ϵs = [randn(n) for i in 1:div(m,2)]

append!(ϵs, -ϵs) # weight mirroring

us = [U(π, ψ + σ.*ϵ) for ϵ in ϵs]

sp = sortperm(us, rev=true)

∇ = sum(w.*ϵs[i] for (w,i) in zip(ws,sp)) / σ

ψ += α.*∇

end

return MvNormal(ψ, σ)

end

Algorithm 10.6. An evolution
strategies method for updating
an isotropic multivariate Gaussian
search distribution with mean ψ

and covariance σ2I over policy pa-
rameterizations for policy π(θ, s).
This implementation also takes a
policy evaluation function U, a step
factor α, and an iteration count
k_max. In each iteration, m/2 param-
eterization samples are drawn and
mirrored, and are then used to es-
timate the search gradient.

Figure 10.8. A demonstration of
the effect mirrored sampling has
on isotropic evolution strategies.
Two-layer neural network policies
were trained on the cart-pole prob-
lem (appendix F.3) using m = 10,
and σ = 0.25 with 6 rollouts per
evaluation. Mirrored sampling sig-
nificantly speeds and stabilizes
learning.

10.7 Summary

• Monte Carlo policy evaluation involves computing expected utility associated
with a policy using a large number of rollouts from states sampled from an
initial state distribution.

• Local searchmethods such as theHooke-Jeevesmethod improve a policy based
on small, local changes.

• Genetic algorithms maintain a population of points in the parameter space,
recombining them in different ways in attempt to drive the population towards
a global optimum.

• The cross entropymethod iteratively improves a search distribution over policy
parameters by re-fitting the distribution to elite samples at each iteration.

• Evolutionary strategies attempt to improve the search distribution using gradi-
ent information from samples from that distribution.

• Isotropic evolutionary strategies make the assumption that the search distribu-
tion is an isotropic Gaussian.

10.8 Exercises

Exercise 10.1. In Monte Carlo policy evaluation, how is the variance of the utility estimate
affected by the number of samples?

Solution: The variance of Monte Carlo policy evaluation is the variance of the mean of m
samples. These samples are assumed independent, and so the variance of the mean is the
variance of a single rollout evaluation divided by the sample size:

1
m

m

∑
i=1

(u(i) − µ)2

where u(i) is the ith rollout utility and µ is their samplemean. The sample variance therefore
decreases with 1/m.

Exercise 10.2. What effect does varying the number of samples m and the number of elite
samples melite have on cross entropy policy search?

Solution: The computational cost per iteration scales linearly with the number of samples.
More samples will better cover the search space, resulting in a better chance of identifying
better elite samples to improve the policy. The number of elite samples also has an effect.
Making all samples elite provides no feedback to the improvement process. Having too
few elite samples can lead to premature convergence to a suboptimal solution.

Exercise 10.3. Consider using evolution strategies with a univariate Gaussian distribution,
θ ∼ N (µ, ν). What is the search gradient with respect to the variance ν? What issue arises
as the variance becomes small?

Solution: The search gradient is the gradient of the log-likelihood:

∂

∂ν
log p(x | µ, ν) =

∂

∂ν
log

1√
2πν

exp

(
− (x− µ)2

2ν

)

=
∂

∂ν

(
−1

2
log(2π)− 1

2
log(ν)− (x− µ)2

2ν

)

= − 1
2ν

+
(x− µ)2

2ν2

0 1 2 3 4 5
0

20

40

ν

∂ ∂
ν

lo
g

p(
1
|0

,ν
)

We find that the gradient goes to infinity as the variance approaches zero. This is a
problem because the variance should be small when the search distribution converges.
Very large gradients can cause simple ascent methods to overshoot optima.

Exercise 10.4. Equation (10.14) defines the objective in terms of a search distribution
θ ∼ N (ψ, Σ). What advantage does this objective have over directly optimizing θ using
the expected utility objective in equation (10.1)?

Solution: The added Gaussian noise around the policy parameters can smooth discontinu-
ities in the original objective, which can lead more reliable optimization.

Exercise 10.5. Which of the methods in this chapter are best suited to reflect the fact that
multiple types of policies could perform well in a given problem?

Solution: The Hooke-Jeeves method improves a single policy parameterization, so it cannot
retain multiple policies. Both the cross entropy method and evolution strategies use search
distributions. In order to successfully represent multiple types of policies, a multi-modal
distribution would have to be used. One common multi-modal distribution is a mixture of
Gaussians. A mixture of Gaussians cannot be fit analytically, but can be reliably fit using
expectaction maximization as demonstrated in example 4.4. Genetic algorithms can retain
multiple policies if the populaiton size is sufficiently large.

Exercise 10.6. Suppose we have a parameterized policy πθ that we would like to optimize
using theHooke-Jeevesmethod. If we initialize our parameter θ = 0 and the utility function
is U(θ) = −3θ2 + 4θ + 1, what is the largest step size α that would still guarantee policy
improvement in the first iteration of the Hooke-Jeeves method?

Solution: The Hooke-Jeeves method evaluates the objective function at the center point ±α

along each coordinate direction. In order to guarantee improvement in the first iteration of
Hooke-Jeeves search, any of the objective function values at the new points must improve
the objective function value. For our policy optimization problem, this means we are
searching for the largest step size α such that either U(θ + α) or U(θ − α) is greater than
U(θ).

Since the underlying utility function is parabolic and concave, the largest step size that
would still lead to improvement is slightly less than the width of the parabola at the current
point. Thus, we compute the point on the parabola opposite the current point, θ′ at which
U(θ′) = U(θ):

U(θ) = −3θ2 + 4θ + 1 = −3(0)2 + 4(0) + 1 = 1

U(θ) = U(θ′)

1 = −3θ′2 + 4θ′ + 1

0 = −3θ′2 + 4θ′ + 0

θ′ =
−4±

√
42 − 4(−3)(0)
2(−3)

=
−4± 4
−6

=
2∓ 2

3
=
{

0, 4
3

}
The point on the parabola opposite the current point is thus θ′ = 4

3 . The distance be-
tween θ and θ′ is 4

3 − 0 = 4
3 . Thus, the maximal step size we can take and still guarantee

improvement in the first iteration is α = 4
3 − ε (where ε is a small number).

Exercise 10.7. Suppose we have a policy parameterized by a single parameter θ. We
take an evolution strategies approach with a search distribution that follows a Bernoulli
distribution p(θ | ψ) = ψθ(1− ψ)1−θ . Compute the log-likelihood gradient ∇ψ log p(θ |
ψ).

Solution: The log-likelihood gradient can be computed as follows:

p(θ | ψ) = ψθ(1− ψ)1−θ

log p(θ | ψ) = log
(

ψθ(1− ψ)1−θ
)

log p(θ | ψ) = θ log ψ + (1− θ) log(1− ψ)

∇ψ log p(θ | ψ) =
d

dψ
[θ log ψ + (1− θ) log(1− ψ)]

∇ψ log p(θ | ψ) =
θ

ψ
− 1− θ

1− ψ

Exercise 10.8. Compute the sample weights for search gradient estimation with rank
shaping given m = 3 samples.

Solution: We first compute the numerator of the first term from equation (10.13), for all i

i = 1 max
(

0, log
(

3
2 + 1

)
− log 1

)
= log 5

2

i = 2 max
(

0, log
(

3
2 + 1

)
− log 2

)
= log 5

4

i = 3 max
(

0, log
(

3
2 + 1

)
− log 3

)
= 0

Now, we compute the weights:

w(1) =
log 5

2

log 5
2 + log 5

4 + 0
− 1

3
= 0.47

w(2) =
log 5

4

log 5
2 + log 5

4 + 0
− 1

3
= −0.14

w(3) =
0

log 5
2 + log 5

4 + 0
− 1

3
= −0.33

